

Functional Programing

Sommersemester 2014, Wintersemester 2015

Author:

Philipp Moers,

Update by: Denis Hirn

Dozent:
Torsten Grust,

ALEXANDER ULRICH

CONTENTS

Abstract

This is just the product of me taking notes on the lecture. Nothing official. If you find mistakes or have got any questions, please feel free to contact me. Cheers!

Contents

1	Intr	roduction	5
2		skell Ramp-Up Function Application	6
3	Valu	ues and Types	7
	3.1	Base Types	7
	3.2	Type Constructors	7
	3.3	Currying	7
	3.4	Defining Values (and thus functions)	8
		3.4.1 Guards	8
		3.4.2 Local Definitions	9
		3.4.3 Lists	9
		3.4.4 Pattern Matching	10
	3.5	Algebraic Data Types	11
4	Тур	pe Classes	16
	4.1	Class Inheritance	16
	4.2	Class Instances	16
		4.2.1 Deriving Class Instances	19
5	Dor	nain-Specific Languages	20
	5.1	Modules	22
	5.2	Generalized Algebraic Data Types (GADTs)	28
	5.3	Shallow embedding of a String Matching DSL	30
		5.3.1 DSL design	30
6	Laz	y Evaluation	35
	6.1	WHNF	35
	6.2	Lazy Evaluation and Bottom	36
7	Infi	nite Lists (Data Structures)	37
	7.1	Example: Newton-Raphson square root approximation	37
	7.2	Example: Tic-Tac-Toe game tree	39
8	Fun	actors	40
	8.1	Examples	40
	8.2	-	41

CONTENTS

9	Applicative	41
	9.1 Applicative instances	42
10	Interlude: Monoid	42
	10.1 Monoid Laws	42
11	Sequencing Functions	43
	11.1 Sequencing partial functions (a -; Maybe b)	43
	11.2 Sequencing exception-generating functions (a -; Exc b)	43
	11.3 Sequencing non-deterministic functions	43
	11.4 Sequencing "stateful" functions	44
	11.5 Sequencing side-effecting functions	44
12	Type class Monad	44
	12.1 Lifting	44
	12.2 do-Notation	45



CONTENTS

Links

Site 2014: http://db.inf.uni-tuebingen.de/teaching/FunctionalProgrammingSS2014.

 $Site\ 2015:\ \texttt{http://db.inf.uni-tuebingen.de/teaching/FunctionalProgrammingWS2015-2016}.$

html Ilias: http://goo.gl/rlqbkK

Literature

• Bird:

Thinking Functionally with Haskell, Cambridge University Press 2014 http://www.cs.ox.ac.uk/publications/books/functional/

- Keller, Chakravarthy: "Learning Haskell", online course in development http://learn.hfm.io/
- Lipovača:

Learn You a Haskell for Great Good No Starch Press 2011, http://learnyouahaskell.com

• O'Sullivan, Steward, Goerzen:
Real World Haskell
O'Reilly 2010
http://book.realworldhaskell.org

• Haskell 2010 Report,

http://www.haskell.org/onlinereport/haskell2010

1 Introduction

Computational model in Functional Programming: **reduction** (replace expression to values) In Functional Programming, expressions are formed by applying functions to values.

- 1. Functions as in math: $x = y \Rightarrow f(x) = f(y)$
- 2. Functions are values (just like numbers, text ...)

	Functional	Imperative
program construction	function application and composition	statement sequencing
execution	reduction (expression evaluation)	state changes
semantics	lambda calculus	complex (denotational)

Example

 $n \in \mathbb{N}, n \geq 2$ is a prime number if the set of non-trivial factors is empty:

```
n \text{ is prime} \Leftrightarrow \{ m \mid m \in \{2, \dots, n-1\}, n \mod m = 0 \} = \emptyset
```

```
1  -- Is n a prime number?
2  isPrime :: Integer -> Bool
3  isPrime n = factors n == []
4  where
5  factors :: Integer -> [Integer]
6  factors n = [ m | m <- [2..n-1], mod n m == 0 ]
7  
8  
9  main :: IO ()
10  main = do
11  let n = 43
12  print (isPrime n)</pre>
```

2 Haskell Ramp-Up

(Read \equiv as "denotes the same value as")

- Apply f to value e: f e (juxtaposition, "apply", binary operator _, Haskell speak: infixL 10 _)
- \Box has max precedence (10): \mathbf{f} e_1 + e_2 \equiv (\mathbf{f} e_1) + e_2
- \square associates to the left: g f e \equiv (g f) e $\neg\neg(g f)$ is a function)
- Function composition:

```
- (g . f) e \equiv g (f e) - (. is something like mathematical | \circ | 'after')
```

Alternative "apply"-operator \$ (lowest precedence, associates to the right, infixR 0 \$):

```
g  $ f $ e \equiv g $ (f $ e) \equiv g (f e)
```

- Prefix application of binary infix operator \otimes : (\otimes) e_1 $e_2 \equiv e_1 \otimes e_2$
- Infix application of binary function f: e_1 'f' $e_2 \equiv f e_1 e_2$:

```
* 1 'elem' [1,2,3] -- (1 \in {1,2,3})  
* n 'mod' m
```

– User defined operators, built from symbols ! # \$ % & * + / $_{\rm i}$ = $_{\rm i}$? \^ |~:.

2.1 Function Application

Any series of identifiers is a function call or, as we often call it, a function application.

```
a b c d
```

This is an application of a function a to three arguments b, c and d.

You may parenthesize function application if you need to.

```
f a b \equiv (f a b) \not\equiv f (a, b)
```

The last one is valid Haskell, but f is a function that takes a pair, (a, b) as an argument.

3 Values and Types

Any Haskell expression e has a type t (e :: t) that is determined at compile time. The **type assignment ::** is either given explicitly or inferred by the compiler.

3.1 Base Types

Type	Description	values
Int	fixed-prec. integer	0, 1, (-42)
Integer	arbitrary prec. integer	10^100
Float, Double	single/double floating point (IEEE)	0.1, 1e02
Char	Unicode character	"x", "\t", "\\", "\8710"
Bool	Boolean	True, False
()	Unit	()

3.2 Type Constructors

- Build new types from existing types
- Let a, b ... denote arbitrary types (type variables)

Type	Description	values
(a, b)	pairs of values of type a, b	(1, True) :: (Int, Bool)
$(a_1, a_2, \ldots a_n)$	n-tuples	
[a]	list of values of type a	[True, False] :: [Bool], []::[a]
Maybe a	optional value of type a	Just 42 :: Maybe Int
		Nothing :: Maybe a
Either a b	choice	Left 'x' :: Either Char b
		Right pi :: Either a Double
IO a	I/O actions that return	print 42 :: IO ()
	a value of type a	
a -> b	functions from a to b	isLetter :: Char -> Bool

3.3 Currying

Currying is the technique of translating the evaluation of a function that takes multiple arguments into evaluating a sequence of functions, each with a single argument.

•
$$Recall$$
: e_1 ++ e_2 \equiv (++) e_1 e_2

• (++)
$$e_1 e_2 \equiv ((++) e_1) e_2$$

• Function application happens one argument at a time.

• Type of n-ary function is
$$a_1 \rightarrow a_2 \rightarrow \dots a_n \rightarrow b$$

• Type fun -> associates to the right,

```
read above type as a_1 \rightarrow (a_2 \rightarrow (\dots (a_n \rightarrow b)))
```

• Enables Partial Application

3.4 Defining Values (and thus functions)

- = binds names to values. Names must not start with A-Z (Haskell style: camelCase)
- Define constant (0-ary function) c. Value of c is value of expression e. c = e
- Define n-ary function f with arguments x_i . f may occur in e. f x_1 x_2 ... x_n = e
- A Haskell program is a set of bindings.
- Good style: give type assignments for top-level (global) bindings:

```
\begin{vmatrix}
\mathbf{f} & :: & \mathbf{a}_1 & \rightarrow & \mathbf{a}_2 & \rightarrow & \mathbf{b} \\
\mathbf{f} & \mathbf{x}_1 & \mathbf{x}_2 & = & \mathbf{e}
\end{vmatrix}
```

3.4.1 Guards

Guards are conditional expressions (something like "switch" in Java). They are a lot more readable and more powerful than if ... then ... else

Guards are introduced by ||:

Guards (q_i) are expressions of type Bool, evaluated top to bottom.

3.4.2 Local Definitions

1. Where bindings: local definitions visible in the entire rhs of a definition.

```
-- Efficient power computation, basic idea: x^2k = (x^2)^k
2
  power :: Double -> Integer -> Double
3
  power x k | k == 1
                       = x
4
             even k
                      = power (x * x) (halve k)
5
             | otherwise = x * power (x * x) (halve k)
6
7
    where
       even n = n \pmod{2} = 0
8
      halve n = n 'div' 2
9
10
  main :: IO ()
  main = print $ power 2 16
```

2. Let expressions: local definitions visible inside one expression.

```
let g_1 = \dots
g_2 = \dots
g_0
in e
```

3.4.3 Lists

• Recursive definitions:

```
    [] is a list (nil), type [] :: [a]
    x:xs is a list, if x :: a, xs :: [a]
    (x is head, xs is tail)
```

```
• Notation: 3:(2:(1:[])) \equiv 3:2:1:[] \equiv [3,2,1] \equiv 3:[2,1]
```

```
• Law: \forall xs :: [a] : (xs \neq [])
head xs : tail xs == xs
```

3.4.4 Pattern Matching

• The idiomatic Haskell way to define a function by cases:

Pattern	Matches If	Bindings in e_r
constant c	$\mathbf{x}_i == \mathbf{c}$	
variable v	always	$v \equiv x_i$
wildcard _	always	
tuple $(p_1, \ldots p_m)$	components of x_i match patterns p	
	$\mathbf{x}_i == []$	
$(p_1: p_2)$	head x_i matches p_1 , tail x_i matches p_2	

```
-- Equivalent definitions of sum (over lists of integers)
1
2
  -- (1) Conditional expression
3
  sum' :: [Integer] -> Integer
  sum' xs = if xs == [] then 0 else head xs + sum' (tail xs)
  -- (2) Guards
  sum'' :: [Integer] -> Integer
8
  sum'' xs | xs == [] = 0
           | otherwise = head xs + sum'' (tail xs)
10
11
   -- (3) Pattern matching
12
  sum''' :: [Integer] -> Integer
13
  sum''' = 0
14
  sum''' (x:xs) = x + sum''' xs
15
16
  main :: IO ()
17
  main = print $ (sum' [1..100], sum'' [1..100], sum'' [1..100])
  -- Finite prefix of a list
1
  take' :: Integer -> [a] -> [a]
2
  take' 0 _ = []
  take' _ [] = []
  take' n (x:xs) = x:take' (n-1) xs
7 | main :: IO ()
8 | main = print $ take' 20 [1,3..]
```

```
-- Mergesort list xs, respecting ordering (<<<)
2
   mergeSort :: (a -> a -> Bool) -> [a] -> [a]
3
                     [] = []
   mergeSort _
                      [x] = [x]
   mergeSort _
5
   mergeSort (<<<) xs = merge (<<<) (mergeSort (<<<) ls)</pre>
                                           (mergeSort (<<<) rs)</pre>
7
     where
8
        (ls, rs) = splitAt (length xs 'div' 2) xs
10
       merge :: (a \rightarrow a \rightarrow Bool) \rightarrow [a] \rightarrow [a] \rightarrow [a]
11
       merge (<<<) []
                             уs
                                         = ys
12
       merge (<<<) xs
                              13
       merge (<<<) 11(x:xs) 12(y:ys)
14
          | x <<< y = x:merge (<<<) xs 12
15
          | otherwise = y:merge (<<<) 11 ys
17
   main :: IO ()
18
   main = print $ mergeSort (>) [1,3..19]
```

3.5 Algebraic Data Types

(also known as Sum-of-Product-Types)

- Recall: [] and (:) are the values constructors for type constructor [a].
- Can define entirely new type T and its constructors K_i :

```
data T a_1 \ a_2 \ \dots \ a_n = K_1 \ b_{11} \ \dots \ b_{1n_1}
K_2 \ b_{21} \ \dots \ b_{2n_2}
K_r \ b_{r1} \ \dots \ b_{r_{n_r}}
```

 b_{ij} types mentioning the type vars $a_1 \dots a_n$

- Defines type constructor T and r value constructors:
 - $K_i :: b_{i_1} \rightarrow b_{i_2} \rightarrow \dots b_{i_n} \rightarrow T a_1 \dots a_n$
- Compare [] :: [a] and (:) :: a -> [a] -> [a]
- Sum Type (n=0, $n_i = 0$)

```
-- Demonstrate algebraic data types: sum type
2
   data Weekday = Mon | Tue | Wed | Thu | Fri | Sat | Sun
     deriving (Eq, Show, Ord, Enum, Bounded)
4
  -- Is this day on a weekend?
6
   weekend :: Weekday -> Bool
   weekend Sat = True
   weekend Sun = True
   weekend _ = False
11
12
   -- The classic rock/paper/scissor game
13
14
   data Move = Rock | Paper | Scissor
15
     deriving (Eq)
17
   data Outcome = Lose | Tie | Win
18
     deriving (Show)
19
20
   -- Outcome of a game round (us vs. them)
21
   outcome :: Move -> Move -> Outcome
22
   outcome Rock
                    Scissor = Win
   outcome Paper Rock
                            = Win
24
   outcome Scissor Paper
                            = Win
25
   outcome us
                    them
26
     | us == them = Tie
27
     | otherwise = Lose
28
29
   main :: IO ()
31
   main = do
32
     print $ Mon == Sun
33
     print $ Thu < Sat</pre>
34
     print [Mon .. Fri]
     print (minBound :: Weekday)
36
     print $ outcome Rock Rock
```

• Add deriving (c, c, ... c) to data declaration to define canonical operations:

c	operations
Eq	equality $(==,/=)$
Show	printing (show)
Ord	ordering $(<, <=, \max)$
Enum	enumeration
Bounded	minBound, maxBound

• Product Types (r=1)

```
-- Demonstrate algebraic data types: product type
2
   data Sequence a = S Int [a]
3
     deriving (Eq, Show)
4
5
6
  fromList :: [a] -> Sequence a
  fromList xs = S (length xs) xs
   (+++) :: Sequence a -> Sequence a
10
   S lx xs +++ S ly ys = S (lx + ly) (xs ++ ys)
11
12
  len :: Sequence a -> Int
13
   len (S lx _) = lx
14
16
  main :: IO ()
17
  main = do
18
     print $ fromList [0..9]
19
    print $ len (fromList ['a'...'m'] +++ fromList ['n'...'z'])
```

• Sum-of-Product-Types

```
data Maybe a = Just a | Nothing
data Either a b = Left a | Right b
data List a = Nil | Cons a (List a)
```

```
-- Our own formulation of cons lists
   data List a = Nil
     | Cons a (List a)
3
   deriving (Show)
4
5
  -- Haskell's builtin type [a] and List a are isomorphic:
6
            toList . fromList = id
7
   -- and fromList . toList = id
  toList :: [a] -> List a
  toList [] = Nil
  toList (x:xs) = Cons x (toList xs)
11
12
13 | fromList :: List a -> [a]
  14
fromList (Cons x xs) = x:fromList xs
  -- The family of well-known list functions (combinators) can be
  -- reformulated for List a
18
  mapList :: (a -> b) -> List a -> List b
19
  mapList f Nil = Nil
20
  mapList f (Cons x xs) = Cons (f x) (mapList f xs)
21
23 | filterList :: (a -> Bool) -> List a -> List a
  filterList p Nil
                                      = Nil
  filterList p (Cons x xs) | p x = Cons x (filterList p xs)
25
                           | otherwise = filterList p xs
26
27
  liftList :: ([a] -> [b]) -> List a -> List b
28
  liftList f = toList . f . fromList
29
  mapList' :: (a -> b) -> List a -> List b
31
  mapList' f = liftList (map f)
32
33
34 | filterList' :: (a -> Bool) -> List a -> List a
  filterList' p = liftList (filter p)
36
  main :: IO ()
  main = print $ fromList $ filterList' (> 3) $ mapList' (+1) $ toList
```

```
-- Abstract syntax tree for arithmetic expressions of literals
  data Exp a = Lit a
              | Add (Exp a) (Exp a)
3
             | Sub (Exp a) (Exp a)
4
             | Mul (Exp a) (Exp a)
    deriving (Show)
6
  ex1 :: Exp Integer
  ex1 = Add (Mul (Lit 5) (Lit 8)) (Lit 2)
9
10
11
  evaluate :: Num a => Exp a -> a
  evaluate (Lit n) = n
  evaluate (Add e1 e2) = evaluate e1 + evaluate e2
  evaluate (Mul e1 e2) = evaluate e1 * evaluate e2
  evaluate (Sub e1 e2) = evaluate e1 - evaluate e2
16
17
18
  main :: IO ()
20 main = print $ evaluate ex1
```

4 Type Classes

A type class C defines a family of type signatures ("methods") which all instances of C must implement.

```
class {	t C} a where {	t f}_1 :: {	t t}_1 ... {	t f}_n :: {	t t}_n
```

The t_i must mention a.

For any f_i the class may provide default implementations.

```
We have f_i :: C a \Rightarrow t_i
```

(read "if a is instance C then f_i has type t_i ").

C a is called class constraint.

Example:

```
class Eq a where

(==) :: a -> a -> Bool

(/=) :: a -> a -> Bool

x == y = not (x /= y)

x /= y = not (x == y)
```

(These are default implementations. To redefine one of them is sufficient.)

4.1 Class Inheritance

- Defining class (c_1 a, c_2 a, ...) => C a where ... makes type class C a subclass of the C_i .
- C a \Rightarrow t implies C_1 a, C_2 a

4.2 Class Instances

If type t implements the methods of class C, t becomes an **instance of** C:

```
instance {f C} t where  {f f}_1 = <\! {\tt def} \ {\tt of} \ {f f}_1 > \\ {\tt  } {\tt
```

(All defs of f_i may be provided, minimal complete definition <u>must</u> be provided.) Class constraint C t is satisfied from now on.

Example:

```
instance Eq Bool where
x == y = x &  y | | (not x &  not y)
```

An instance definition for type constructor t may formulate class constraints for its argument types a, b, ...: instance (C_1 a, C_2 a, ...) => C t where

4.2 Class Instances 4 TYPE CLASSES

```
import Data.Maybe
import Data.Tuple
-- The classic rock/paper/scissor game
data Outcome = Lose | Tie | Win
instance Eq Outcome where
 Lose == Lose = True
 Tie == Tie = True
 Win == Win = True
     == _ = False
instance Enum Outcome where
  fromEnum Lose = 0
  fromEnum Tie = 1
  fromEnum Win = 2
  toEnum 0 = Lose
  toEnum 1 = Tie
  toEnum 2 = Win
instance Show Outcome where
  show Lose = "Lose"
  show Tie = "Tie"
  show Win = "Win"
instance Ord Outcome where
 Lose <= Lose = True
 Lose <= Tie = True
 Lose <= Win = True
 Tie <= Tie = True
 Tie <= Win = True
  Win <= Win = True
  _ <= _ = False
instance Bounded Outcome where
 minBound = Lose
 maxBound = Win
data Move = Rock | Paper | Scissor
instance Eq Move where
 Rock == Rock = True
```

4.2 Class Instances 4 TYPE CLASSES

```
Paper == Paper = True
 Scissor == Scissor = True
  _ == _ = False
-- Lookup table defining a consistent mapping between Move and Int
table :: [(Move, Int)]
table = [(Rock, 0), (Paper, 1), (Scissor, 2)]
instance Enum Move where
 fromEnum o = fromJust $ lookup o table
 toEnum n = fromJust $ lookup n $ map swap table
instance Show Move where
 show Rock = "Rock"
 show Paper = "Paper"
 show Scissor = "Scissor"
instance Ord Move where
 Rock <= Rock = True
 Rock <= Paper = True
 Rock <= Scissor = True
 Paper <= Paper = True
 Paper <= Scissor = True
 Scissor <= Scissor = True
        <= _ = False
instance Bounded Move where
 minBound = Rock
 maxBound = Scissor
outcome :: Move -> Move -> Outcome
outcome Paper Rock = Win
outcome Scissor Paper = Win
outcome us them
  | us == them = Tie
  | otherwise = Lose
main :: IO ()
main = print $ outcome Paper Paper
```

4.2 Class Instances 4 TYPE CLASSES

4.2.1 Deriving Class Instances

Automatically make user-defined data types (data ...) instances of classes $C_i \in \{ Eq, Ord, Enum, Bounded, Show, Read \}:$

```
data T a_1 a_1 ... a_n = ...
                       1 ...
2
       deriving (C_1, C_2, \ldots)
   -- Use deriving the obtain the standard interpretation for the type
1
   → classes
   -- Eq, Ord, Enum, Bounded, Show, Read
   -- The classic rock/paper/scissor game
   data Outcome = Lose | Tie | Win
     deriving (Eq, Ord, Enum, Bounded, Show, Read)
6
  data Move = Rock | Paper | Scissor
8
     deriving (Eq, Enum, Read)
                                   -- Ord, Show defined below;
9
                                   -- Bounded makes no sense
10
   instance Show Move where
                 = "rock"
     show Rock
12
     show Paper = "paper"
13
     show Scissor = "scissor"
14
15
   instance Ord Move where
                                     -- NB: non-conventional,
16
     Rock
            <= Rock
                        = True
                                    -- encodes game rules
17
     Rock
             <= Paper
                         = True
18
     Paper
             <= Paper
                       = True
19
             <= Scissor = True
20
     Scissor <= Scissor = True
21
     Scissor <= Rock
                        = True
22
             <= _
                         = False
23
   outcome :: Move -> Move -> Outcome
   outcome m1 m2 | m1 == m2 = Tie
26
                  | m1 < m2 = Lose
27
                  | otherwise = Win
28
29
  main :: IO ()
30
  main = do
31
32
     print $ outcome Paper Scissor
     print $ [Rock, Paper, Scissor]
33
     print $ (read "Scissor" :: Move)
34
```

5 Domain-Specific Languages

a.k.a. DSLs

- "small" languages designed to easily and directly express the concepts/idioms of a specific domain. Not Turing-complete in general.
- Examples:

Domain	DSLs
OS automation	shell scripts, OSX Automater
Typesetting	Ŀ₽ŢĘX
Queries	SQL
Game Scripting	Unreal Script, Lua
Parsing	Yacc, Bison, ANTLR

- Functional Languages make good hosts for **embedded DSLs**:
 - algebraic data types (e.g. to model ASTs)
 - higher-order functions (abstraction, control constructs)
 - lightweight syntax (layout / whitespace, non-alphabetic ids)

Example: An embedded DSL for integer sets:

(1) DSL as library of functions, implementation details exposed.

```
import Data.List (nub)
   -- A library of functions on integer sets,
  -- implementation fully exposed
   type IntegerSet = [Integer] -- unsorted, duplicates allowed
6
   empty :: IntegerSet
7
   empty = []
   insert :: Integer -> IntegerSet -> IntegerSet
   insert x xs = x:xs
11
12
   delete :: Integer -> IntegerSet -> IntegerSet
13
   delete x xs = filter (/= x) xs
14
15
   (∈) :: Integer -> IntegerSet -> Bool
16
   x \in xs = elem x xs
17
20
   -- "Extending" the library, accessing the exposed
21
   -- implementation. Now we're doomed to stick the
22
   -- list-based representation...
23
24
   (□) :: IntegerSet -> IntegerSet -> Bool
   xs \subseteq ys = all (\x -> x \in ys) xs
26
27
   card :: IntegerSet -> Int
28
   card xs = length (nub xs)
30
31
32
  s1, s2 :: IntegerSet
33
   s1 = insert 1 (insert 2 (insert 3 empty))
34
  s2 = foldr insert empty [1..10]
35
  prog :: Bool
37
  prog = s1 \subseteq s2
38
39
  main :: IO ()
40
  main = print $ prog
```

5.1 Modules

• Group of related definitions (values, types) in a single file (named "M.hs" / "M.lhs"):

```
module M where
type Predicate a = a -> Bool
id :: a -> a
id x = x
```

- \bullet Hierarchy: module A.B.C.M in file A/B/C/M.hs
- Access definitions in other module M: import M

• Explicit export lists hide all other definitions:

```
module M (id) where

...

-- type Predicate a not exported
```

• Abstract data types:

export algebraic data types, but <u>not</u> its constructors:

```
module M (Rose, leaf) where
data Rose a = Node a [Rose a]
leaf :: a -> Rose a [Rose a]
leaf x = Node x []
```

- Export constructors:

```
module M (Rose(Node), leaf) where

...

-- or export all constructors:
module M (Rose(..), leaf) where
```

- Instance definitions (including deriving) are exported with their type.
- Qualified imports to partition name space:

```
import qualified M

...

-- use M.foobar syntax

t :: M.Rose Char

t = M.leaf 'x'
```

• Partially import module:

```
-- only import nub and reverse
import Data.List (nub, reverse)

-- import whole module but without otherwise
import Prelude hiding (otherwise)
otherwise :: Bool
otherwise = False -- harhar
```

• Pass imported modules to importer of own module:

```
module M (..., module Data.List, ...) where
import Data.List (nub)

import qualified M
M.nub
```

```
module SetLanguageShallowCard (IntegerSet,
                                     empty,
2
                                     insert,
3
                                     delete,
                                     member,
5
                                     card) where
6
7
   data IntegerSet = IS (Integer -> Bool) Int -- characteristic function +
   \rightarrow cardinality
   -- constructors
10
   empty :: IntegerSet
11
   empty = IS (\_ -> False)
12
              0
13
   -- empty = IS (const False) 0
14
   {-
   insert :: IntegerSet -> Integer -> IntegerSet
16
   insert \ xs@(IS \ f \ c) \ x = IS \ (\y \rightarrow x == y \ // \ f \ y)
17
                                 (if member xs x then c else c + 1)
18
   -7
19
   delete :: IntegerSet -> Integer -> IntegerSet
20
   delete xs \circ (IS f c) x = IS (y -> y /= x && f y)
21
                                (if member xs x then c - 1 else c)
22
23
   -- observer
24
  member :: IntegerSet -> Integer -> Bool
25
  member (IS f _) x = f x
26
   -- member (IS f _) = f
27
28
   card :: IntegerSet -> Int
29
  card (IS _ c) = c
```

```
--import SetLanguage
  import SetLanguageShallow
2
  --import SetLanguageShallowCard
   -- impossible
5
   -- (SetLanguage/SetLanguageShallow do not export
      data constructor IS)
   {-
  union :: IntegerSet -> IntegerSet -> IntegerSet
   union (IS xs) (IS ys) = IS (xs ++ ys)
11
12
  union :: IntegerSet -> IntegerSet -> IntegerSet
13
   union (IS f) (IS g) = IS (y \rightarrow f y | g y)
14
15
   -}
17
  set12 :: IntegerSet
18
  set12 = (((empty 'insert' 3) 'insert' 2) 'delete' 3) 'insert' 1
19
20
  main :: IO ()
21
  main = print $ set12 'member' 3
```

(2) DSL as library of functions, abstract data type (module).

• Shallow DSL embedding:

Semantics of DSL operations directly expressed in terms of host language value (e.g. list or characteristic function)

- constructors (empty, insert, delete) perform the work, harder to add
- observers (member) trivial

• Deep DSL embedding:

DSL operations build an abstract syntax tree (AST) that represents applications and arguments

- constructors merely build the AST, very easy to add
- observers interpret (traverse) the AST and perform the work

```
module SetLanguageDeepCard (IntegerSet(Empty, Insert, Delete),
                                member,
2
                                card) where
3
   -- constructors
5
6
   data IntegerSet =
7
       Empty
8
     | Insert IntegerSet Integer
9
     | Delete IntegerSet Integer
10
     deriving (Show)
11
12
   -- TODO: build a suitable Show instance for IntegerSet
13
14
   -- observers
15
   member :: IntegerSet -> Integer -> Bool
17
                  _ = False
   member Empty
18
   member (Insert xs x) y = x == y || member xs y
19
   member (Delete xs x) y = x /= y && member xs y
20
21
   card :: IntegerSet -> Int
22
   card Empty
                                     = 0
   card (Insert xs x) | member xs x = card xs
                       | otherwise = card xs + 1
25
   card (Delete xs x) | member xs x = card xs - 1
26
                       | otherwise = card xs
27
   import SetLanguageDeepCard
1
2
   set12 :: IntegerSet
3
   set12 = (((Empty 'Insert' 3) 'Insert' 2) 'Delete' 3) 'Insert' 1
4
5
   main :: IO ()
   main = do
    print $ set12
8
    print $ set12 'member' 3
9
    print $ card set12
10
```

```
module ExprDeepNum (Expr(..),
                        eval) where
2
3
   -- constructors
5
   data Expr =
6
       Val Integer
7
     | Add Expr Expr
8
     | Mul Expr Expr
     | Sub Expr Expr
     deriving (Show)
11
12
   instance Num Expr where
13
     fromInteger n = Val n
14
     e1 + e2 = Add e1 e2
15
     e1 - e2 = Sub e1 e2
     e1 * e2 = Mul e1 e2
^{17}
18
     abs e = undefined
19
     signum e = undefined
20
21
22
   -- observer
23
24
   eval :: Expr -> Integer
25
  eval (Val n) = n
26
  eval (Add e1 e2) = eval e1 + eval e2
27
  eval (Mul e1 e2) = eval e1 * eval e2
  eval (Sub e1 e2) = eval e1 - eval e2
```

```
import ExprDeepNum
2
   -- e1 = 8 * 7 - 14
   e1 :: Expr
   e1 = Sub (Mul (Val 8) (Val 7)) (Val 14)
6
   e2 :: Expr
   e2 = 8 * 7 - 14
  main :: IO ()
10
  main = do
11
     print $ eval e1
12
     print $ negate e2
13
     print $ eval e2
14
```

5.2 Generalized Algebraic Data Types (GADTs)

Idea:

- Encode the type of a DSL expression (here: Integer or Bool) in its **Haskell representation type**.
- Use Haskell's type checker to ensure at compile time that only well-typed DSL expressions are built.

```
Language extensions:
```

```
\{-\# LANGUAGE GADTs \#-\}
```

• Define new parameterized type T, its constructors k_i and their type signatures:

```
 \begin{vmatrix} \text{data T a}_1 & \text{a}_2 & \dots & \text{a}_n & \text{where} \\ \text{k}_1 & :: & \text{b}_{11} & -> & \dots & -> & \text{b}_1 n_1 & -> & \text{T } \textbf{t}_{11} & \textbf{t}_{1n} \\ & & & & & & & & & \\ \text{4} & & & & & & & & \\ \text{k}_r & :: & \text{b}_{r1} & -> & \dots & -> & \text{b}_r n_r & -> & \text{T } \textbf{t}_{r1} & \textbf{t}_{rn} \end{vmatrix}
```

```
{-# LANGUAGE GADTs #-}
2
  module ExprDeepGADTTyped (Expr(..),
                              eval) where
4
5
  -- Expr a: an expression that, if evaluated, will yield a value of type
6
   \hookrightarrow a
  data Expr a where
            :: Integer
    ValI
                                               -> Expr Integer
    ValB :: Bool
                                              -> Expr Bool
10
            :: Expr Integer -> Expr Integer -> Expr Integer
11
            :: Expr Bool -> Expr Bool
                                             -> Expr Bool
12
    EqZero :: Expr Integer
                                              -> Expr Bool
13
           :: Expr Bool -> Expr a -> Expr a -> Expr a
14
16
   instance Show (Expr a) where
17
     show (ValI n)
                       = show n
18
     show (ValB b)
                      = show b
19
     show (Add e1 e2) = show e1 ++ " + " ++ show e2
20
     show (And e1 e2) = show e1 ++ " \wedge " ++ show e2
21
     show (EqZero e) = show e ++ " == 0"
22
     show (If p e1 e2) = "if " ++ show p ++ " then " ++ show e1 ++ " else "
23
     \rightarrow ++ show e2
24
25
   -- NB: this is *typed* evaluation of expressions:
26
  eval :: Expr a -> a
27
  eval (ValI n)
                    = n
  eval (ValB b)
                    = b
29
  eval (Add e1 e2) = eval e1 + eval e2
30
  eval (And e1 e2) = eval e1 && eval e2
31
  eval (EqZero e) = eval e == 0
32
eval (If p e1 e2) = if eval p then eval e1 else eval e2
```

```
import ExprDeepGADTTyped
2
  e1 :: Expr Integer
  e1 = If (EqZero (Add (ValI 0) (ValI 0))) (ValI 42) (ValI 43)
5
   -- e2 yields compile-time type error
6
7
  e2 :: Expr Bool
  e2 = EqZero (ValB True)
  main :: IO ()
11
  main = do
12
     print $ e1
13
     print $ eval e1
14
```

Shallow embedding of a String Matching DSL 5.3

• Pattern:

- 1. Given a String, a pattern returns the list of matches. Match failure? Returns the empty list.
- 2. A match consists of a value (e.g. the matched characters, tokens, parse trees) and the residual String left to match.

Thus: type Pattern a = String -> [(a, String)]

5.3.1 DSL design

Pattern		DSL function
match lit. char	"x"	lit :: Char -> Pattern Char
match empty string	ϵ	empty :: a -> Pattern a
fail always	Ø	fail :: Pattern a
alternative		alt :: Pattern a -> Pattern a -> Pattern a
sequence		seq :: (a -> b -> c) ->
		Pattern a -> Pattern b -> Pattern c
repetition	*	rep :: Pattern a -> Pattern [a]

```
module PatternMatching (Pattern,
                         module Prelude,
                        lit, empty, fail,
                         alt, seq, rep, rep1,
```

```
alts, seqs, lits, app) where
import Prelude hiding (seq, fail)
-- Given a string, a pattern returns the (possibly empty) list of
-- possible matches. A match consists of a match value (e.g., matched
-- the matched character(s), token, or parse tree) and the residual
\rightarrow string
-- left to match:
type Pattern a = String -> [(a, String)]
-- BASIC PATTERNS
-- match character c, returning the matched character
lit :: Char -> Pattern Char
lit _c []
lit c (x:xs) \mid c == x = [(c, xs)]
              | otherwise = []
-- match the empty string, return v
empty :: a -> Pattern a
empty v xs = [(v, xs)]
-- fail always (yields empty list of matches)
fail :: Pattern a
fail _ = []
-- COMBINE PATTERNS
-- match p or q
alt :: Pattern a -> Pattern a -> Pattern a
alt p q xs = p xs ++ q xs
-- match p and q in sequence (use f to combine match values)
seq :: (a -> b -> c) -> Pattern a -> Pattern b -> Pattern c
seq f p q xs = concat (map (\((v1, xs1) ->
                         map (\(v2, xs2) -> (f v1 v2, xs2))
                             (q xs1))
                         (p xs))
-- An alternative (more consise and readable) implementation of seq
```

```
-- based on list comprehension syntax:
 -- seq f p q xs = [ (f v1 v2, xs2) | (v1, xs1) <- p xs, (v2, xs2) <- q
  \rightarrow xs1 ]
-- match p repeatedly (including 0 times)
rep :: Pattern a -> Pattern [a]
rep p = alt (seq (:) p (rep p)) (empty [])
-- match p repeatedly, but at least once
rep1 :: Pattern a -> Pattern [a]
rep1 p = seq (:) p (rep p)
-- CONVENIENCE
-- build "or" choice pattern from a list of alternatives
alts :: [Pattern a] -> Pattern a
alts = foldr alt fail
-- build "and" sequence pattern from a list of patterns
seqs :: [Pattern a] -> Pattern [a]
seqs = foldr (seq (:)) (empty [])
-- match a string (= sequence of characters)
lits :: String -> Pattern String
lits cs = seqs [ lit c | c <- cs ]</pre>
-- apply function f to match value (for match post-processing)
app :: (a -> b) -> Pattern a -> Pattern b
app f p xs = [ (f v1, xs1) | (v1, xs1) \leftarrow p xs ]
{-
-- Using rep leads to longest match first:
rep p xs
      = alt (seq (:) p (rep p)) (empty []) xs
      = seq (:) p (rep p) xs ++ empty [] xs
      = seq (:) p (rep p) xs ++ [([], xs)]
       = [(v1:v2, xs2) | (v1,xs1) \leftarrow p xs, (v2,xs2) \leftarrow rep p xs1] ++ [([], xs2) \leftarrow rep p xs2] ++ [([], xs2) \leftarrow rep xs2] ++ [([], xs2] \leftarrow rep
  \rightarrow xs)]
```

```
rep P'a' "aab"
  = [(v1:v2, xs2) | (v1,xs1) \leftarrow P'a' \text{ "aab"}, (v2,xs2) \leftarrow rep P'a' xs1]

→ ++ [([], "aab")]

  = [(v1:v2, xs2) | (v1,xs1) \leftarrow [('a',"ab")], (v2,xs2) \leftarrow rep P'a' xs1
\rightarrow ] ++ [([], "aab")]
  = [('a':v2,xs2) | (v2,xs2) \leftarrow rep P'a' "ab"] ++ [([], "aab")]
-}
```

```
import Prelude ()
   import PatternMatching
  fortytwo :: Pattern (Char, Char)
5
   fortytwo = seq (,) (lit '4') (lit '2')
6
   digit :: Pattern Char
   digit = lit '0' 'alt' lit '1' 'alt' lit '2' 'alt' lit '3' 'alt'
9
           lit '4' 'alt' lit '5' 'alt' lit '6' 'alt' lit '7' 'alt'
           lit '8' 'alt' lit '9'
11
12
   number :: Pattern String
13
   number = rep digit
14
   smiley :: Pattern String
16
   smiley = seq (:) (alt (lit ':') (lit ';'))
17
                     (seq (:) (lit '-')
18
                               (seq (:) (alt (lit '(') (lit ')'))
19
                                        (empty [])))
20
21
   main :: IO ()
22
   main = do
23
     print $ fortytwo "42 rules"
24
     print $ rep digit "42 rules"
25
     print $ smiley ";-)"
26
```

```
import Prelude ()
import PatternMatching
-- Make use of the fact that the pattern matching DSL is *embedded*
-- into Haskell: define new functions (abstractions) that combine
-- simple patterns
```

```
-- Example:
-- Match a fully parenthesized arithmetic expression over integers,
-- e.g. ((4*10)+2)
-- Variant 1: return list of matched characters
digit :: Pattern Char
digit = alts [ lit d | d <- ['0'...'9'] ]
number :: Pattern String
number = rep1 digit
op :: Pattern String
op = alts [ lits o | o <- ["+", "-", "*", "/"] ]
expr :: Pattern String
expr = alts [ number, app concat (seqs [lits "(", expr, op, expr, lits
→ ")"]) ]
-- Variant 2: return a simple AST for the matched expression
data Expr a =
    Num a
  | Op (Expr a) String (Expr a)
  deriving (Show)
number' :: Pattern (Expr Integer)
number' = app (Num . read) (rep1 digit)
expr' :: Pattern (Expr Integer)
expr' = alts [ number', seq (\ (e1,(o,(e2,\_))) \rightarrow Op \ e1 \ o \ e2)
                             (lit '(') (seq (,)
                                       expr' (seq (,)
                                             op (seq (,)
                                                expr' (lit ')'))))
             ]
main :: IO ()
main = do
  print $ rep1 digit "1234.56"
  print $ lits "abc" "abcdef"
  print $ expr "((4*10)+2)"
  print $ expr' "((4*10)+2)"
```

6 Lazy Evaluation

To execute a program, Haskell reduces expressions to values. Haskell uses normal order reduction to select the next expression to reduce:

- 1. The outermost reducible expression (redex) is reduced first
- 2. \Rightarrow Function applications are reduced first before their arguments.
- 3. If no further redex is found, the expression is in normal form and reduction terminates.

```
fst :: (a, b) -> a
fst (x, y) = x
sqr :: Num a => a -> a
sqr x = x * x

-- ->: reduces to
fst (sqr (1+3), sqr 2) -> sqr (1 + 3) [fst]
-> (1+3) * (1+3) [sqr]
-> 4 * 4 [+/+]
-> 16 [*]
```

Haskell avoids the duplication of work through graph reduction. Expression are shared (referenced more than once) instead of duplicated.

Lazy evaluation: normal order reduction and sharing.

6.1 WHNF

An expression e is in weak head normal form (WHNF) if it is of the following form

1. v (where v is an atomic value

```
Integer, Bool, Char, ...
)
2. c e1 e2 ... en
  (where c is an n-ary constructor, like
  (:)
)
3. f e1 e2 ... em
  (where f is an n-ary function, m < n)</pre>
```

Haskell reduces values to WHNF only (stop criterium for reduction) unless we request reduction to normal form (e.g. when printing results).

Example expressions in WHNF:

```
42 --1.
(sqr 2, sqr 4) --2. (,)
f x : map f xs --2. (:)
Just (40+2) --2. Just
(* (40+2)) --3. * binär
(\x -> 40+2) --3. unary function w/o args
```

6.2 Lazy Evaluation and Bottom

 $f x1 \dots xi-1 bottom xi+1 \dots xn = bottom$

Some Haskell expressions have the value bottom. Examples:

```
error, undefined, bomb
```

. Lazy evaluation admits functions that return a non-bottom value even if they receive bottom as argument (also: non-strict functions). N-ary function f is strict in its i-th argument, if

```
Examples:

• const :: a -> b -> a --strict in first, non-strict in second argument

• (&&) :: Bool -> Bool -> Bool -- dito
```

If a function pattern matches on an argument, Haskell semantics define it to be strict in that argument.

Example:

```
data T = T Int
f :: T -> Int
f (T x) = 42

f undefined -> undefined
f (T undefined) -> 42
```

7 Infinite Lists (Data Structures)

One consequence of lazy evaluation: programs can handle infinite Lists as long as any run will inspect only a finite prefix of such a list. Enables a modular programming style:

- 1. generator functions produce an infinite number of solutions / approximations / ...
- 2. test functions select one (or finite number of) solutions from this infinite list.

7.1 Example: Newton-Raphson square root approximation

Iteratively approximate the square root of x:

1.
$$a_0 = \frac{x}{2}$$

$$2. \ a_{i+1} = \frac{a_i + \frac{x}{a_i}}{2}$$

```
-- Demonstrate modular program construction through laziness:
  -- value generation (here: iterate) and consume/test (here: within)
  -- can be implemented separately.
  -- Can replace test (within \rightarrow relative) without modifying the
   \rightarrow generator.
  -- See John Hughes, "Why Functional Programming Matters", Section 4.1
5
  import Prelude hiding (iterate)
  -- [x, f x, f (f x), f (f (f x)), \ldots]
  iterate :: (a -> a) -> a -> [a]
  iterate f x = x : iterate f (f x)
  -- Consume list until two adjacent elements are
10
   -- 1. within eps of each other
11
  -- 2. differ by a factor less than eps
12
  within :: (Ord a, Num a) => a -> [a] -> a
13
   within eps (x1:x2:xs) | abs (x1 - x2) \le eps = x2
14
                          | otherwise
                                                  = within eps (x2:xs)
16
   relative :: (Ord a, Fractional a) => a -> [a] -> a
17
   relative eps (x1:x2:xs) | abs (x1/x2 - 1) \le eps = x2
18
                            otherwise
                                                      = relative eps (x2:xs)
19
   -- Square root of x using the Newton-Raphson algorithm:
20
       a0 = x / 2
21
        ai+1 = (ai + x / ai) / 2
22
   -- Why does this work? If the approximations ai converge to some
23
   -- limit a, then:
24
      a = (a + x / a) / 2
25
   -- 2a = a + x / a
26
       a = x / a
27
   -- a^2 = x
28
       a = \sqrt{x}
29
   sqroot :: Double -> Double
30
   sqroot eps x = within eps (iterate next a0)
31
                  relative
32
    where
33
       -- initial approximation
34
       a0 :: Double
35
       a0 = x / 2
36
       -- find next ai+1, given ai
37
       next :: Double -> Double
38
       next a = (a + x / a) / 2
39
40
  main :: IO ()
41
  main = print $ sqroot 0.001 81
```

7.2 Example: Tic-Tac-Toe game tree

Build the (potentially huge) tree of possible moves for the Tic-Tac-Toe board game. Evaluate promise of game position. Plan:

1. Find representation of game position (board + player next up)

```
|1|2|3| next: x
|4|5|6| square #6: open
|0|x|0| spare #9: occupied by player 0
```

- 2. Provide pretty-printing for game positions.
- 3. Define initial position and possible moves:

```
moves :: Position -> [Position]
```

4. Evaluate a given position:

```
static :: Position -> Int
(1 / -1: x/0 won the game, 0 draw)
```

5. Build a game tree of positions:

```
gameTree :: Position -> Tree Position
```

- 6. Rather than simple static evaluation, now evaluate positions based on possible game futures. ⇒ in game tree, perform evaluation bottom up.
- 7. Optimization $(\alpha \beta \text{ algorithm})$

Code: tic-tac-toe.hs

8 Functors

Type class Functor embodies the application of a function to the elements (or: inside) of a structure, which leaving structure (or: outside) alone.

8.1 Examples

```
map :: (a \rightarrow b) \rightarrow [a] \rightarrow [b]
   mapTree :: (a -> b) -> Tree a -> Tree b
3
   --Note: f is a type constructor that receives exactly one argument
   -- (Functor is also called a constructor class).
   class Functor f where
6
      fmap :: (a \rightarrow b) \rightarrow f a \rightarrow f b
   --Examples:
9
   instance Functor [] where
10
      fmap = map
11
12
   instance Functor Tree where
13
      fmap = mapTree
14
15
   instance Functor Maybe where
16
      fmap f (Just x) = Just (f x)
17
      fmap f Nothing = Nothing
18
```

Type constructors can be partially applied. Uses prefix notation:

```
a \rightarrow b \equiv (->) a b
   (a, b) \equiv (,) a b
   [a] \equiv [] a
   -- Examples (defines type constructors):
   type Flagged = (,) Bool
6
   type Indexed (->) Int
   --MayFail e a: computation may yield value a or fail with error e
   type MayFail e = Either e
10
   instance Functor (Either e) where
11
      fmap f (Left e) = Left e
12
      fmap f (Right x) = Right (f x)
13
14
   instance Functor Flagged where
15
   --fmap :: (a \rightarrow b) \rightarrow (Bool, a) \rightarrow (Bool, b)
16
      fmap f (b, x) = (b, f x)
17
18
   instance Functor Indexed where
19
      fmap f g = f . g
20
^{21}
   --Functor Laws
22
   fmap id \equiv id
   fmap (f . g) = fmap f . fmap g
```

8.2 Kinds - Types for Types

kind	describes	example	
*	types	Int, Bool, (Bool, Int), [Char],	
* -> *	unary type constructor	Maybe, []	
* -> * -> *	binary type constructor	Either, (,), (->)	

9 Applicative

```
--Compare:

($) :: (a -> b) -> a -> b

fmap :: Functor f => (a -> b) -> f a -> f b --<$>

(**>) :: Applicative f => f (a -> b) -> f a -> f b

Read <*> as (ap)ply, "tiefighter".
```

```
class Functor f => Applicative f where
pure :: a -> f a
(<*>) :: f (a -> b) -> f a -> f b

--Make any Applicative f a Functor:
fmap g x = pure g <*> x
```

Applicative embodies

- 1. function application on the level of (contained) values.
- 2. <u>combination</u> on the level of structures.

9.1 Applicative instances

```
instance Applicative Maybe where
      pure x = Just x
      Just f \ll Just x = Just (f x)
3
      _ <*> _ = Nothing
4
5
   instance Monoid c => Applicative ((,) c) where
6
      pure x = (mempty, x)
7
      (c_1, f) \iff (c_2, x) = (c_1 \text{ 'mappend'} c_2, f x)
   instance Applicative [] where
10
      pure x = [x]
11
      fs \ll xs = [f x | f \ll fs, x \ll xs]
12
```

10 Interlude: Monoid

Type class Monoid a represents combinable values of type a:

```
class Monoid a where

mempty :: a -- empty, neutral element for mappend

mappend :: a -> a -> a -- combination

mconcat :: [a] -> a

Examples: (0, +), (1, \cdot), (\text{true}, \wedge), (\text{false}, \vee), (\{\}, \cup), ([], (++))
```

10.1 Monoid Laws

```
mempty 'mappend'xs = xs
xs 'mappend'(ya 'mappend'zs) = (xs 'mappend'ys) 'mappend'zs''
mconcat xs = foldr mempty mappend xs
```

11 Sequencing Functions

```
1  (.) :: (b -> c) -> (a -> b) -> (a -> c)
2  f . g = \x -> f (g x)
3  4  (>=>) :: (a -> b) -> (b -> c) -> (a -> c)
5  f >=> g = g . f
6  f1 >=> f2 >=> ... >=> fn -- composes the fi in left-to-right order
8  -- (>=>, id) forms a monoid.
```

11.1 Sequencing partial functions (a -; Maybe b)

Return Nothing as soon as first function in pipeline returns Nothing

11.2 Sequencing exception-generating functions (a -; Exc b)

```
type Exc b = Either Error b
type Error = String

Exceptions are propagated once occurred:

(>=>) :: (a -> Exc b) -> (b -> Exc c) -> (a -> Exc c)

f >=> g = \x -> case f x of

Left err -> Left err
Right y -> g y
```

11.3 Sequencing non-deterministic functions

```
(a -> NonDet b)
--Non-determinism: any answer in a list of answers:
type NonDet b = [b]
--Take all possible answers into account:
(>=>) :: (a -> NonDet b) -> (b -> NonDet c) -> (a -> NonDet c)
f >=> g = \x -> concat [g y | y <- f x]</pre>
```

11.4 Sequencing "stateful" functions

```
1  a -> State -> (State, b)
2  --State Transformer:
3  type ST b = State -> (State, b)
4  --Stateful function: a -> ST b
5  (>=>) :: (a -> ST b) -> (b -> ST c) -> (a -> ST c)
6  f >=> g = \ x s0 -> let (s1, y) = f x s0
7  let (s2, z) = g y s1 -- <=> in g y s1
8  in (s2, z)
```

11.5 Sequencing side-effecting functions

```
a -> World -> (World, b)

--Side effects: functions consume current world

--return new world along with result:

type World = ... -- abstract (defined by Haskell runtime)

type IO b = World -> (World, b)

--Value of type IO b: an action that

--1. when performed has a side-effect on the world, and then

--2. returns a value of type b.
```

Haskell build-ins:

```
print :: Show a => a -> IO ()
putStrLn :: String -> IO ()
getLine :: IO String
readFile :: FilePath -> IO String
```

12 Type class Monad

```
class Applicative m => Monad m where
return :: a -> m a
(>>=) :: m a -> (a -> m b) -> m b -- bind
```

12.1 Lifting

Monad	Lifting (return x)
Maybe	Just x
Exc	Right x
NonDet	[x]
ST	\s -> (s , x)

```
--Make Maybe an instance of Monad:
instance Monad Maybe where
return = Just
Nothing >>= g = Nothing
Just y >>= g = g y

--Kleisli composition
(>=>) :: Monad m => (a -> m b) -> (b -> m c) -> (a -> m c)
f >=> g = \x -> f x >>= g
```

Let m be a Monad. Then m is also an Applicative:

```
pure = return
v <*> v = u >>= \f -> v >>= \x -> return (f x)

($) :: (a -> b) -> a -> b
(<$>) :: Functor m => (a -> b) -> m a -> m b
(<*>) :: Applicative m => m (a -> b) -> m a -> m b
(=<<) :: Monad m => (a -> m b) -> m a -> m b -- (=<<) = flip (>>=)
```

12.2 do-Notation

```
1     --(e : expression of type m a, es: sequence of ;-separated expression):
2     do { e } \equiv e
3     do {x <- e; es} \equiv e >>= \x -> do {es}
4     do {e; es} \equiv e >>= \_ -> do {es}
5     do {let v = x; es} = let v = x in do {es}
```